

Linear perturbation of the Schwarzschild metric

Preliminaries

Kernels for parallel evaluation

In the following we will represent tensor fields by arrays, i.e., nested lists. To reduce the time needed for simplification, we simplify their components in parallel. The following command launches several parallel kernels, corresponding to the number of available CPU cores.

```
In[1]:= LaunchKernels[];
```

Assumptions on coordinates

To aid with the simplification, we specify the ranges of the coordinates and parameters we use.

```
In[2]:= $Assumptions = {r > 0, -1 ≤ χ ≤ 1, 0 ≤ φ ≤ 2 Pi, m ∈ Integers, l ∈ Integers};
```

```
In[3]:= ParallelEvaluate[$Assumptions = {r > 0, -1 ≤ χ ≤ 1, 0 ≤ φ ≤ 2 Pi, m ∈ Integers, l ∈ Integers}];
```

Coordinates

The coordinates we use. Note that we use $\chi = \cos \vartheta$ instead of the azimuth angle ϑ .

```
In[4]:= co = {t, r, χ, φ}
```

```
Out[4]= {t, r, χ, φ}
```

Simplification to linear order

Since we are working with linear perturbations, which we define with respect to a perturbation parameter ϵ , we discard all terms of higher than linear order, and then simplify.

```
In[5]:= linsimp[x_] := FullSimplify[Normal[Series[x, {ε, 0, 1}]]]
```

For convenience, we also define a function which simplifies all tensor components in parallel.

```
In[6]:= tensimp[x_] := ParallelMap[linsimp, x, {TensorRank[x]}]
```

Formatting

We will use a number of functions to express the metric perturbations. In order to shorten the expressions, we define a number of shorthand notations for these functions and their derivatives.

```
In[7]:= Format[Y[l_, m_, χ_, φ_]] := Subscript["Y", StringJoin@@(ToString/@{l, m})]
```

```
In[8]:= Format[Derivative[0, 0, c_, 0][Y[l_, m_, χ_, φ_]] := Subsuperscript["Y", StringJoin@@(ToString/@{l, m}), StringRepeat["'", c]]
```

```

In[9]:= Format[sp1[t_, r_] := Subscript["S", "tt"]
In[10]:= Format[sp2[t_, r_] := Subscript["S", "tr"]
In[11]:= Format[sp3[t_, r_] := Subscript["S", "rr"]
In[12]:= Format[vp1[t_, r_] := Subscript["V", "t"]
In[13]:= Format[vp2[t_, r_] := Subscript["V", "r"]
In[14]:= Format[va1[t_, r_] := Subscript["A", "t"]
In[15]:= Format[va2[t_, r_] := Subscript["A", "r"]
In[16]:= Format[tt[t_, r_] := "T"
In[17]:= Format[tp[t_, r_] := "P"
In[18]:= Format[ta[t_, r_] := "E"
In[19]:= Format[Derivative[ct_, cr_][sp1][t_, r_] := Subscript["S", StringJoin["tt,", StringRepeat["t", ct], StringRepeat["r", cr]]]
In[20]:= Format[Derivative[ct_, cr_][sp2][t_, r_] := Subscript["S", StringJoin["tr,", StringRepeat["t", ct], StringRepeat["r", cr]]]
In[21]:= Format[Derivative[ct_, cr_][sp3][t_, r_] := Subscript["S", StringJoin["rr,", StringRepeat["t", ct], StringRepeat["r", cr]]]
In[22]:= Format[Derivative[ct_, cr_][vp1][t_, r_] := Subscript["V", StringJoin["t,", StringRepeat["t", ct], StringRepeat["r", cr]]]
In[23]:= Format[Derivative[ct_, cr_][vp2][t_, r_] := Subscript["V", StringJoin["r,", StringRepeat["t", ct], StringRepeat["r", cr]]]
In[24]:= Format[Derivative[ct_, cr_][va1][t_, r_] := Subscript["A", StringJoin["t,", StringRepeat["t", ct], StringRepeat["r", cr]]]
In[25]:= Format[Derivative[ct_, cr_][va2][t_, r_] := Subscript["A", StringJoin["r,", StringRepeat["t", ct], StringRepeat["r", cr]]]
In[26]:= Format[Derivative[ct_, cr_][tt][t_, r_] := Subscript["T", StringJoin["", StringRepeat["t", ct], StringRepeat["r", cr]]]
In[27]:= Format[Derivative[ct_, cr_][tp][t_, r_] := Subscript["P", StringJoin["", StringRepeat["t", ct], StringRepeat["r", cr]]]
In[28]:= Format[Derivative[ct_, cr_][ta][t_, r_] := Subscript["E", StringJoin["", StringRepeat["t", ct], StringRepeat["r", cr]]]
In[29]:= Format[v[r_] := "v"
In[30]:= Format[λ[r_] := "λ"
In[31]:= Format[Derivative[c_][v][r_] := Superscript["v", StringRepeat["", c]]
In[32]:= Format[Derivative[c_][λ][r_] := Superscript["λ", StringRepeat["", c]]
In[33]:= Format[ξt[t_, r_] := Subscript["ξ", "t"]
In[34]:= Format[ξr[t_, r_] := Subscript["ξ", "r"]
In[35]:= Format[ξp[t_, r_] := "Π"
In[36]:= Format[ξa[t_, r_] := "Ξ"
In[37]:= Format[Derivative[ct_, cr_][ξt][t_, r_] := Subscript["ξ", StringJoin["t,", StringRepeat["t", ct], StringRepeat["r", cr]]]
In[38]:= Format[Derivative[ct_, cr_][ξr][t_, r_] := Subscript["ξ", StringJoin["r,", StringRepeat["t", ct], StringRepeat["r", cr]]]
In[39]:= Format[Derivative[ct_, cr_][ξp][t_, r_] := Subscript["Π", StringJoin["", StringRepeat["t", ct], StringRepeat["r", cr]]]
In[40]:= Format[Derivative[ct_, cr_][ξa][t_, r_] := Subscript["Ξ", StringJoin["", StringRepeat["t", ct], StringRepeat["r", cr]]]
In[41]:= Format[fsp1[r_] := Subscript["Ŝ", "tt"]

```

```

In[42]:= Format[fsp2[r_]] := Subscript["Ŝ", "tr"]
In[43]:= Format[fsp3[r_]] := Subscript["Ŝ", "rr"]
In[44]:= Format[fva1[r_]] := Subscript["Â", "t"]
In[45]:= Format[fva2[r_]] := Subscript["Â", "r"]
In[46]:= Format[ftt[r_]] := "î"
In[47]:= Format[Derivative[c_][fsp1][r_]] := Subsuperscript["Ŝ", "tt", StringRepeat["", c]]
In[48]:= Format[Derivative[c_][fsp2][r_]] := Subsuperscript["Ŝ", "tr", StringRepeat["", c]]
In[49]:= Format[Derivative[c_][fsp3][r_]] := Subsuperscript["Ŝ", "rr", StringRepeat["", c]]
In[50]:= Format[Derivative[c_][fva1][r_]] := Subsuperscript["Â", "t", StringRepeat["", c]]
In[51]:= Format[Derivative[c_][fva2][r_]] := Subsuperscript["Â", "r", StringRepeat["", c]]
In[52]:= Format[Derivative[c_][ftt][r_]] := Superscript["î", StringRepeat["", c]]
In[53]:= Format[fax[r_]] := "ℱ"
In[54]:= Format[Derivative[c_][fax][r_]] := Superscript["ℱ", StringRepeat["", c]]

```

Two-dimensional geometry on the sphere

Metric

We define the metric γ_{ab} in the angular coordinates (χ, φ) we use.

```
In[55]:= MatrixForm[Met2DD = DiagonalMatrix[{1/(1 - χ^2), 1 - χ^2}]]
```

Out[55]//MatrixForm=

$$\begin{pmatrix} \frac{1}{1-\chi^2} & 0 \\ 0 & 1-\chi^2 \end{pmatrix}$$

Inverse metric

The inverse metric is immediate.

```
In[56]:= MatrixForm[Met2UU = Inverse[Met2DD]]
```

Out[56]//MatrixForm=

$$\begin{pmatrix} 1-\chi^2 & 0 \\ 0 & \frac{1}{1-\chi^2} \end{pmatrix}$$

Antisymmetric tensor

We define the totally antisymmetric tensor on the sphere with a negative sign, since the coordinate frame in the (χ, φ) has the opposite orientation compared to the standard (ϑ, φ) coordinates, and we use the latter orientation.

```
In[57]:= MatrixForm[Eps2DD = -LeviCivitaTensor[2] Sqrt[Det[Met2DD]]]
```

```
Out[57]//MatrixForm=
```

$$\begin{pmatrix} 0 & -1 \\ 1 & 0 \end{pmatrix}$$

Christoffel symbols

Finally, the Levi-Civita connection of the metric γ_{ab} .

```
In[58]:= TableForm[Gam2UDD = Table[Sum[Met2UU[[a, d]] (D[Met2DD[[d, c]], co[[b + 2]]) + D[Met2DD[[b, d]], co[[c + 2]]) - D[Met2DD[[b, c]], co[[d + 2]]] / 2, {d, 2}], {a, 2}, {b, 2}, {c, 2}]]
```

```
Out[58]//TableForm=
```

$$\begin{array}{cc} \frac{\chi}{1-\chi^2} & 0 \\ 0 & \chi(1-\chi^2) \\ 0 & -\frac{\chi}{1-\chi^2} \\ -\frac{\chi}{1-\chi^2} & 0 \end{array}$$

Tensor harmonics

Derivatives of spherical harmonics

Here we implement the rules for derivatives of the spherical harmonics Y_{lm} by hand, since we use rather non-standard coordinates. These are defined by the Laplace equation and the usual angular dependence.

```
In[59]:= yru = {Derivative[0, 0, 2, 0][Y][l_, m_, x_, phi] -> (2 Derivative[0, 0, 1, 0][Y][l, m, x, phi] x - (l(l+1) - m^2 / (1-x^2)) Y[l, m, x, phi]) / (1-x^2),
  Derivative[0, 0, a_, b_ /; b > 0][Y][l_, m_, x_, phi] -> Derivative[0, 0, a, 0][Y][l, m, x, phi] (I m)^b};
```

This is their effect.

```
In[60]:= D[Y[l, m, x, phi], {x, 2}] /. yru
```

```
Out[60]=
```

$$\frac{-\left(\left(l(l+1) - \frac{m^2}{1-x^2}\right) Y_{lm}\right) + 2\chi Y'_{lm}}{1-x^2}$$

```
In[61]:= D[Y[l, m, x, phi], phi] /. yru
```

```
Out[61]=
```

$$i m Y_{lm}$$

We can also define them explicitly.

```
In[62]:= YY[l_, m_, x_, phi] := Sqrt[(2 l + 1) (l - m)! / (l + m)! / 4 / Pi] LegendreP[l, m, x] Exp[I m phi]
```

These are the first few harmonics.

```
In[63]:= TableForm[Table[YY[l, m, χ, φ], {l, 0, 2}, {m, -l, l}]]
```

```
Out[63]//TableForm=
```

$$\frac{1}{2\sqrt{\pi}}$$

$$\frac{1}{2} e^{-i\varphi} \sqrt{\frac{3}{2\pi}} \sqrt{1-\chi^2} \quad \frac{1}{2} \sqrt{\frac{3}{\pi}} \chi \quad -\frac{1}{2} e^{i\varphi} \sqrt{\frac{3}{2\pi}} \sqrt{1-\chi^2}$$

$$\frac{1}{4} e^{-2i\varphi} \sqrt{\frac{15}{2\pi}} (1-\chi^2) \quad \frac{1}{2} e^{-i\varphi} \sqrt{\frac{15}{2\pi}} \chi \sqrt{1-\chi^2} \quad \frac{1}{4} \sqrt{\frac{5}{\pi}} (-1+3\chi^2) \quad -\frac{1}{2} e^{i\varphi} \sqrt{\frac{15}{2\pi}} \chi \sqrt{1-\chi^2} \quad -\frac{1}{4} e^{2i\varphi} \sqrt{\frac{15}{2\pi}} (-1+\chi^2)$$

Scalar harmonic

The scalar harmonics are simply the spherical harmonics themselves. They have polar parity.

```
In[64]:= YS[l_, m_, χ_, φ_] := Y[l, m, χ, φ]
```

The first non-vanishing scalar is the monopole $l = 0$.

```
In[65]:= YS[0, 0, χ, φ] /. Y → YY
```

```
Out[65]=
```

$$\frac{1}{2\sqrt{\pi}}$$

Vector harmonics

For the vector harmonics, we have two contributions. The first one has polar parity.

```
In[66]:= YVP[l_, m_, χ_, φ_] := D[YS[l, m, χ, φ], #] & /@ {χ, φ} /. yru
```

```
In[67]:= YVP[l, m, χ, φ]
```

```
Out[67]=
```

$$\{Y'_{lm}, i m Y_{lm}\}$$

The second part has axial parity, which can be seen from the appearance of the Levi-Civita tensor.

```
In[68]:= YVA[l_, m_, χ_, φ_] := Eps2DD.Met2UU.YVP[l, m, χ, φ]
```

```
In[69]:= YVA[l, m, χ, φ]
```

```
Out[69]=
```

$$\left\{ -\frac{i m Y_{lm}}{1-\chi^2}, (1-\chi^2) Y'_{lm} \right\}$$

For the vectors, the monopole vanishes, and the first non-vanishing term is the dipole $l = 1$.

```
In[70]:= TableForm[Map[MatrixForm, Table[FullSimplify[YVP[l, m, χ, φ] /. Y → YY], {l, 0, 1}, {m, -l, l}], {2}]]
```

```
Out[70]//TableForm=
```

$$\begin{pmatrix} 0 \\ 0 \end{pmatrix}$$

$$\begin{pmatrix} -\frac{\sqrt{3} e^{-i\varphi} \chi}{2\sqrt{2\pi-2\pi\chi^2}} \\ -\frac{1}{2} i e^{-i\varphi} \sqrt{\frac{3}{2\pi}} \sqrt{1-\chi^2} \end{pmatrix} \begin{pmatrix} \frac{\sqrt{3}}{2} \\ 0 \end{pmatrix} \begin{pmatrix} \frac{\sqrt{3} e^{i\varphi} \chi}{2\sqrt{2\pi-2\pi\chi^2}} \\ -\frac{1}{2} i e^{i\varphi} \sqrt{\frac{3}{2\pi}} \sqrt{1-\chi^2} \end{pmatrix}$$

```
In[71]:= TableForm[Map[MatrixForm, Table[FullSimplify[YVA[l, m, x, phi] /. Y -> YY], {l, 0, 1}, {m, -l, l}], {2}]]
```

```
Out[71]//TableForm=
```

$$\begin{pmatrix} 0 \\ 0 \end{pmatrix} \begin{pmatrix} \frac{i \sqrt{3} e^{-i\phi}}{2 \sqrt{2\pi-2\pi x^2}} \\ -\frac{1}{2} e^{-i\phi} \sqrt{\frac{3}{2\pi}} x \sqrt{1-x^2} \end{pmatrix} \begin{pmatrix} 0 \\ -\frac{1}{2} \sqrt{\frac{3}{\pi}} (-1+x^2) \end{pmatrix} \begin{pmatrix} \frac{i \sqrt{3} e^{i\phi}}{2 \sqrt{2\pi-2\pi x^2}} \\ \frac{1}{2} e^{i\phi} \sqrt{\frac{3}{2\pi}} x \sqrt{1-x^2} \end{pmatrix}$$

Tensor harmonics

Finally, we come to the tensor harmonics. The first one is just the diagonal trace component. It has polar parity.

```
In[72]:= YTT[l_, m_, x_, phi_] := YS[l, m, x, phi] Met2DD
```

```
In[73]:= MatrixForm[YTT[l, m, x, phi]]
```

```
Out[73]//MatrixForm=
```

$$\begin{pmatrix} \frac{Y_{lm}}{1-x^2} & 0 \\ 0 & (1-x^2) Y'_{lm} \end{pmatrix}$$

Also the second covariant derivative gives a term with polar parity.

```
In[74]:= YTP[l_, m_, x_, phi_] := Module[{a, b, c}, Table[D[YS[l, m, x, phi], co[[a+2]], co[[b+2]]] - Sum[D[YS[l, m, x, phi], co[[c+2]]] * Gam2UDD[c, a, b], {c, 2}], {a, 2}, {b, 2}] /. yru]
```

```
In[75]:= MatrixForm[FullSimplify[YTP[l, m, x, phi]]]
```

```
Out[75]//MatrixForm=
```

$$\begin{pmatrix} \frac{(m^2+1)(1+l)(-1+x^2) Y_{lm} - X(-1+x^2) Y'_{lm}}{(-1+x^2)^2} & i m \left(\frac{X Y_{lm}}{1-x^2} + Y'_{lm} \right) \\ i m \left(\frac{X Y_{lm}}{1-x^2} + Y'_{lm} \right) & -m^2 Y_{lm} + X(-1+x^2) Y'_{lm} \end{pmatrix}$$

The third and last term has axial parity.

```
In[76]:= YTA[l_, m_, x_, phi_] := (# + Transpose[#]) / 2 &[Eps2DD.Met2UU.YTP[l, m, x, phi]]
```

```
In[77]:= MatrixForm[FullSimplify[YTA[l, m, x, phi]]]
```

```
Out[77]//MatrixForm=
```

$$\begin{pmatrix} \frac{i m (-X Y_{lm} + (-1+x^2) Y'_{lm})}{(-1+x^2)^2} & \frac{1}{2} \left(-l(1+l) - \frac{2m^2}{-1+x^2} \right) Y_{lm} + X Y'_{lm} \\ \frac{1}{2} \left(-l(1+l) - \frac{2m^2}{-1+x^2} \right) Y_{lm} + X Y'_{lm} & -i m (-X Y_{lm} + (-1+x^2) Y'_{lm}) \end{pmatrix}$$

Note that the axial tensor has vanishing trace by definition:

```
In[78]:= Tr[Met2UU.YTA[l, m, x, phi]]
```

```
Out[78]=
```

0

The trace of the polar tensor, however, is given by the Laplacian:

```
In[79]:= FullSimplify[Tr[Met2UU.YTP[l, m, x, phi]]]
```

```
Out[79]=
```

$-l(1+l) Y_{lm}$

Often one therefore also uses a trace-free polar tensor instead:

```
In[80]:= FullSimplify[Tr[Met2UU.(YTP[l, m, χ, φ]+l(l+1)YTT[l, m, χ, φ]/2)]]
```

```
Out[80]=
```

0

For the two trace-free tensor harmonics, the lowest order is the quadrupole $l=2$.

```
In[81]:= TableForm[Map[MatrixForm, Table[FullSimplify[YTP[l, m, χ, φ]+l(l+1)YTT[l, m, χ, φ]/2 /. Y → YY], {l, 0, 2}, {m, -l, l}], {2}]]
```

```
Out[81]//TableForm=
```

$$\begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix} \quad \begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix} \quad \begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix}$$

$$\begin{pmatrix} -\frac{e^{-2i\varphi} \sqrt{\frac{15}{2\pi}} (1+\chi^2)}{4(-1+\chi^2)} & \frac{1}{2} i e^{-2i\varphi} \sqrt{\frac{15}{2\pi}} \chi \\ \frac{1}{2} i e^{-2i\varphi} \sqrt{\frac{15}{2\pi}} \chi & \frac{1}{4} e^{-2i\varphi} \sqrt{\frac{15}{2\pi}} (-1+\chi^4) \end{pmatrix} \quad \begin{pmatrix} -\frac{\sqrt{15} e^{i\varphi} \chi}{2\sqrt{2\pi-2\pi\chi^2}} & -\frac{1}{2} i e^{-i\varphi} \sqrt{\frac{15}{2\pi}} \sqrt{1-\chi^2} \\ -\frac{1}{2} i e^{-i\varphi} \sqrt{\frac{15}{2\pi}} \sqrt{1-\chi^2} & \frac{1}{2} e^{-i\varphi} \sqrt{\frac{15}{2\pi}} \chi (1-\chi^2)^{3/2} \end{pmatrix} \quad \begin{pmatrix} \frac{3\sqrt{\frac{5}{\pi}}}{4} & 0 \\ 0 & -\frac{3}{4} \sqrt{\frac{5}{\pi}} (-1+\chi^2)^2 \end{pmatrix}$$

$$\begin{pmatrix} \frac{\sqrt{15} e^{i\varphi} \chi}{2\sqrt{2\pi-2\pi\chi^2}} & -\frac{1}{2} i e^{i\varphi} \sqrt{\frac{15}{2\pi}} \sqrt{1-\chi^2} \\ -\frac{1}{2} i e^{i\varphi} \sqrt{\frac{15}{2\pi}} \sqrt{1-\chi^2} & -\frac{1}{2} e^{i\varphi} \sqrt{\frac{15}{2\pi}} \chi (1-\chi^2)^{3/2} \end{pmatrix} \quad \begin{pmatrix} -\frac{e^{2i\varphi} \sqrt{\frac{15}{2\pi}} (1+\chi^2)}{4(-1+\chi^2)} & -\frac{1}{2} i e^{2i\varphi} \sqrt{\frac{15}{2\pi}} \chi \\ -\frac{1}{2} i e^{2i\varphi} \sqrt{\frac{15}{2\pi}} \chi & \frac{1}{4} e^{2i\varphi} \sqrt{\frac{15}{2\pi}} (-1+\chi^4) \end{pmatrix}$$

```
In[82]:= TableForm[Map[MatrixForm, Table[FullSimplify[YTA[l, m, χ, φ] /. Y → YY], {l, 0, 2}, {m, -l, l}], {2}]]
```

```
Out[82]//TableForm=
```

$$\begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix} \quad \begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix} \quad \begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix}$$

$$\begin{pmatrix} \frac{i e^{-2i\varphi} \sqrt{\frac{15}{2\pi}} \chi}{2(-1+\chi^2)} & \frac{1}{4} e^{-2i\varphi} \sqrt{\frac{15}{2\pi}} (1+\chi^2) \\ \frac{1}{4} e^{-2i\varphi} \sqrt{\frac{15}{2\pi}} (1+\chi^2) & -\frac{1}{2} i e^{-2i\varphi} \sqrt{\frac{15}{2\pi}} \chi (-1+\chi^2) \end{pmatrix} \quad \begin{pmatrix} \frac{i \sqrt{15} e^{i\varphi}}{2\sqrt{2\pi-2\pi\chi^2}} & -\frac{1}{2} e^{-i\varphi} \sqrt{\frac{15}{2\pi}} \chi \sqrt{1-\chi^2} \\ -\frac{1}{2} e^{-i\varphi} \sqrt{\frac{15}{2\pi}} \chi \sqrt{1-\chi^2} & -\frac{1}{2} \sqrt{\frac{15}{2\pi}} (1-\chi^2)^{3/2} (i \cos[\varphi] + \sin[\varphi]) \end{pmatrix} \quad \begin{pmatrix} 0 & -\frac{3}{4} \sqrt{\frac{5}{\pi}} (-1+\chi^2) \\ -\frac{3}{4} \sqrt{\frac{5}{\pi}} (-1+\chi^2) & 0 \end{pmatrix}$$

$$\begin{pmatrix} \frac{i \sqrt{15} e^{i\varphi}}{2\sqrt{2\pi-2\pi\chi^2}} & \frac{1}{2} e^{i\varphi} \sqrt{\frac{15}{2\pi}} \chi \sqrt{1-\chi^2} \\ \frac{1}{2} e^{i\varphi} \sqrt{\frac{15}{2\pi}} \chi \sqrt{1-\chi^2} & -\frac{1}{2} i e^{i\varphi} \sqrt{\frac{15}{2\pi}} (1-\chi^2)^{3/2} \end{pmatrix} \quad \begin{pmatrix} \frac{i e^{2i\varphi} \sqrt{\frac{15}{2\pi}}}{2-2\chi^2} \\ \frac{1}{4} e^{2i\varphi} \sqrt{\frac{15}{2\pi}} \end{pmatrix}$$

Calculate Laplacian

It is instructive to calculate the Laplacian of the tensor harmonics, which is one possibility to define them (along with the action of the angular momentum operators).

```
In[83]:= FullSimplify[Sum[Met2UU[[a, b]] (D[YS[l, m, χ, φ], co[[a+2]], co[[b+2]] - Sum[D[YS[l, m, χ, φ], co[[c+2]]] × Gam2UDD[[c, a, b]], {c, 2}]), {a, 2}, {b, 2}]/YS[l, m, χ, φ] /. yru]
```

```
Out[83]=
```

$-l(l+1)$

Spacetime geometry

Background metric

We proceed with the spacetime geometry. This is the unperturbed background metric. Here we assume a spherically symmetric, static metric, but we leave the g_{tt} and g_{rr} components as free functions of the radial coordinate for now.

In[84]:= **MatrixForm**[**Met4DD0** = **DiagonalMatrix**[[{-Exp[2 v[r]], Exp[2 λ[r]], r^2/(1-χ^2), r^2(1-χ^2)}]]

Out[84]//MatrixForm=

$$\begin{pmatrix} -e^{2v} & 0 & 0 & 0 \\ 0 & e^{2\lambda} & 0 & 0 \\ 0 & 0 & \frac{r^2}{1-\chi^2} & 0 \\ 0 & 0 & 0 & r^2(1-\chi^2) \end{pmatrix}$$

Metric perturbation

We then define the metric perturbation.

In[85]:= **MatrixForm**[**Met4DD1** = **tensimp**[**ArrayFlatten**[[{{{sp1[t, r] × YS[l, m, χ, φ]}, {{sp2[t, r] × YS[l, m, χ, φ]}}, {vp1[t, r] × YVP[l, m, χ, φ] + va1[t, r] × YVA[l, m, χ, φ]}}, {{{sp2[t, r] × YS[l, m, χ, φ]}}, {{sp3[t, r] × YS[l, m, χ, φ]}}, {vp2[t, r] × YVP[l, m, χ, φ] + va2[t, r] × YVA[l, m, χ, φ]}}, {**Transpose**[[vp1[t, r] × YVP[l, m, χ, φ] + va1[t, r] × YVA[l, m, χ, φ]}], **Transpose**[[vp2[t, r] × YVP[l, m, χ, φ] + va2[t, r] × YVA[l, m, χ, φ]}], tt[t, r] × YTT[l, m, χ, φ] + tp[t, r] × YTP[l, m, χ, φ] + ta[t, r] × YTA[l, m, χ, φ]}] // . yru]]

Out[85]//MatrixForm=

$$\begin{pmatrix} S_{tt} Y_{lm} & S_{tr} Y_{lm} & \frac{i m A_t Y_{lm}}{-1+\chi^2} + V_t Y'_{lm} & i m V_t Y_{lm} - (-1+\chi^2) A_t Y'_{lm} \\ S_{tr} Y_{lm} & S_{rr} Y_{lm} & \frac{i m A_r Y_{lm}}{-1+\chi^2} + V_r Y'_{lm} & i m V_r Y_{lm} - (-1+\chi^2) A_r Y'_{lm} \\ \frac{i m A_t Y_{lm}}{-1+\chi^2} + V_t Y'_{lm} & \frac{i m A_r Y_{lm}}{-1+\chi^2} + V_r Y'_{lm} & \frac{(-i m \chi E + (m^2 + l(l+1)(-1+\chi^2)) P - (-1+\chi^2) T) Y_{lm} - (-1+\chi^2) (-i m E + \chi P) Y'_{lm}}{(-1+\chi^2)^2} & -\frac{((2 m^2 + l(l+1)(-1+\chi^2)) E + 2 i m \chi P) Y_{lm}}{2(-1+\chi^2)} + (\chi E + i m P) Y'_{lm} \\ i m V_t Y_{lm} - (-1+\chi^2) A_t Y'_{lm} & i m V_r Y_{lm} - (-1+\chi^2) A_r Y'_{lm} & -\frac{((2 m^2 + l(l+1)(-1+\chi^2)) E + 2 i m \chi P) Y_{lm}}{2(-1+\chi^2)} + (\chi E + i m P) Y'_{lm} & (i m \chi E - m^2 P - (-1+\chi^2) T) Y_{lm} + (-1+\chi^2) (-i m E + \chi P) Y'_{lm} \end{pmatrix}$$

Full metric

The full metric is given as the sum of the background metric and the perturbation. Note that we introduce the perturbation parameter ϵ here, which will be used to select only linear perturbations by performing a Taylor expansion.

In[86]:= **Met4DD** = **Met4DD0** + ϵ **Met4DD1**;

Inverse metric

This is the inverse metric. Note that only terms up to linear order in ϵ appear, since we have discarded all higher order terms in the simplification function.

In[87]:= **MatrixForm**[**Met4UU** = **tensimp**[**Inverse**[**Met4DD**]]]

Out[87]//MatrixForm=

$$\begin{pmatrix} -e^{-4v} (e^{2v} + \epsilon S_{tt} Y_{lm}) & e^{-2(\lambda+v)} \epsilon S_{tr} Y_{lm} & \frac{e^{-2v} \epsilon (-i m A_t Y_{lm} - (-1+\chi^2) V_t Y'_{lm})}{r^2} & \frac{e^{-2v} \epsilon (-i m V_t Y_{lm} + (-1+\chi^2) A_t Y'_{lm})}{r^2 (-1+\chi^2)} \\ e^{-2(\lambda+v)} \epsilon S_{tr} Y_{lm} & e^{-4\lambda} (e^{2\lambda} - \epsilon S_{rr} Y_{lm}) & \frac{e^{-2\lambda} \epsilon (i m A_r Y_{lm} + (-1+\chi^2) V_r Y'_{lm})}{r^2} & \frac{e^{-2\lambda} \epsilon (i m V_r Y_{lm} - (-1+\chi^2) A_r Y'_{lm})}{r^2 (-1+\chi^2)} \\ \frac{e^{-2v} \epsilon (-i m A_t Y_{lm} - (-1+\chi^2) V_t Y'_{lm})}{r^2} & \frac{e^{-2\lambda} \epsilon (i m A_r Y_{lm} + (-1+\chi^2) V_r Y'_{lm})}{r^2} & \frac{-r^2 (-1+\chi^2) - \epsilon (-i m \chi E + (m^2 + l(l+1)(-1+\chi^2)) P - (-1+\chi^2) T) Y_{lm} + \epsilon (-1+\chi^2) (-i m E + \chi P) Y'_{lm}}{r^4} & -\frac{\epsilon \left(-\frac{((2 m^2 + l(l+1)(-1+\chi^2)) E + 2 i m \chi P) Y_{lm}}{2(-1+\chi^2)} + (\chi E + i m P) Y'_{lm} \right)}{r^4} \\ \frac{e^{-2v} \epsilon (-i m V_t Y_{lm} + (-1+\chi^2) A_t Y'_{lm})}{r^2 (-1+\chi^2)} & \frac{e^{-2\lambda} \epsilon (i m V_r Y_{lm} - (-1+\chi^2) A_r Y'_{lm})}{r^2 (-1+\chi^2)} & \frac{\epsilon \left(-\frac{((2 m^2 + l(l+1)(-1+\chi^2)) E + 2 i m \chi P) Y_{lm}}{2(-1+\chi^2)} + (\chi E + i m P) Y'_{lm} \right)}{r^4} & \frac{-r^2 (-1+\chi^2) + \epsilon (-i m \chi E + m^2 P + (-1+\chi^2) T) Y_{lm} - \epsilon (-1+\chi^2) (-i m E + \chi P) Y'_{lm}}{r^4 (-1+\chi^2)^2} \end{pmatrix}$$

Christoffel symbols

Next, we calculate the Christoffel symbols.

```
In[88]:= Gam4UDD = tensimp[Table[Sum[Met4UU[[a, d]] (D[Met4DD[[d, c]], co[[b]] + D[Met4DD[[b, d]], co[[c]] - D[Met4DD[[b, c]], co[[d]]) / 2, {d, 4}], {a, 4}, {b, 4}, {c, 4}] /. yru];
```

It is helpful to note also the background of the Christoffel symbols. It can be used to calculate the covariant derivative of the metric perturbation, which then yields the perturbation of the Riemann tensor.

```
In[89]:= TableForm[FullSimplify[Gam4UDD /.  $\epsilon \rightarrow 0$ ]]
```

```
Out[89]//TableForm=
```

0	v'	0	0
v'	0	0	0
0	0	0	0
0	0	0	0
$e^{-2\lambda+2\nu} v'$	0	0	0
0	λ'	0	0
0	0	$\frac{e^{-2\lambda} r}{-1+\chi^2}$	0
0	0	0	$e^{-2\lambda} r (-1+\chi^2)$
0	0	0	0
0	0	$\frac{1}{r}$	0
0	$\frac{1}{r}$	$\frac{\chi}{1-\chi^2}$	0
0	0	0	$\chi - \chi^3$
0	0	0	0
0	0	0	$\frac{1}{r}$
0	0	0	$\frac{\chi}{-1+\chi^2}$
0	$\frac{1}{r}$	$\frac{\chi}{-1+\chi^2}$	0

Riemann tensor

This gives us the perturbed Riemann tensor.

```
In[90]:= Riem4UDDD = tensimp[Table[D[Gam4UDD[[a, b, d]], co[[c]] - D[Gam4UDD[[a, b, c]], co[[d]] + Sum[Gam4UDD[[a, e, c]]  $\times$  Gam4UDD[[e, b, d]] - Gam4UDD[[a, e, d]]  $\times$  Gam4UDD[[e, b, c]], {e, 4}], {a, 4}, {b, 4}, {c, 4}, {d, 4}] /. yru];
```

Also the background value of the Riemann tensor is easily obtained.

Check spherical symmetry

Killing vector fields on the sphere

We can check the expansion of the perturbations into tensor spherical harmonics by calculating their behavior under rotations. For this purpose, we need the rotation generating vector fields on the sphere, in the relevant coordinates.

```
In[95]:= kvf2 = tensimp[{{Sin[φ], Cos[φ]/Tan[φ]}, {-Cos[φ], Sin[φ]/Tan[φ]}, {0, -1}}.DiagonalMatrix[{-Sin[φ], 1}]/. {φ → ArcCos[x]}
```

```
Out[95]=
```

$$\left\{ \left\{ -\sqrt{1-x^2} \sin[\varphi], \frac{x \cos[\varphi]}{\sqrt{1-x^2}} \right\}, \left\{ \sqrt{1-x^2} \cos[\varphi], \frac{x \sin[\varphi]}{\sqrt{1-x^2}} \right\}, \{0, -1\} \right\}$$

We see that they indeed satisfy the rotation algebra.

```
In[96]:= tensimp[Table[Sum[kvf2[[i, b]] * D[kvf2[[j, a]], co[[b + 2]]] - kvf2[[j, b]] * D[kvf2[[i, a]], co[[b + 2]]], {b, 2}] - Sum[Signature[{i, j, k}] * kvf2[[k, a]], {k, 3}], {i, 3}, {j, 3}, {a, 2}]]
```

```
Out[96]=
```

$$\{\{0, 0\}, \{0, 0\}, \{0, 0\}\}, \{\{0, 0\}, \{0, 0\}, \{0, 0\}\}, \{\{0, 0\}, \{0, 0\}, \{0, 0\}\}$$

Killing vector fields on spacetime

We lift the vector fields to the full spacetime manifold by noticing that they do not modify the time and radial coordinates.

```
In[97]:= kvf4 = PadLeft[#, 4] & /@ kvf2
```

```
Out[97]=
```

$$\left\{ \left\{ 0, 0, -\sqrt{1-x^2} \sin[\varphi], \frac{x \cos[\varphi]}{\sqrt{1-x^2}} \right\}, \left\{ 0, 0, \sqrt{1-x^2} \cos[\varphi], \frac{x \sin[\varphi]}{\sqrt{1-x^2}} \right\}, \{0, 0, 0, -1\} \right\}$$

Symmetry of two-dimensional geometry

By construction, the two-dimensional metric and its Levi-Civita tensor on the sphere are invariant under rotations.

```
In[98]:= tensimp[Table[Sum[kvf2[[i, c]] * D[#, a, b], co[[c + 2]]] + #[[c, b]] * D[kvf2[[i, c]], co[[a + 2]]] + #[[a, c]] * D[kvf2[[i, c]], co[[b + 2]]], {c, 2}], {i, 3}, {a, 2}, {b, 2}] & /@ {Met2DD, Eps2DD}]]
```

```
Out[98]=
```

$$\{\{\{0, 0\}, \{0, 0\}\}, \{\{0, 0\}, \{0, 0\}\}, \{\{0, 0\}, \{0, 0\}\}\}, \{\{\{0, 0\}, \{0, 0\}\}, \{\{0, 0\}, \{0, 0\}\}, \{\{0, 0\}, \{0, 0\}\}\}$$

Symmetry of four-dimensional background geometry

Also the spacetime metric is invariant under rotations by construction.

```
In[99]:= tensimp[Table[Sum[kvf4[[i, c]] * D[Met4DD0[[a, b]], co[[c]]] + Met4DD0[[c, b]] * D[kvf4[[i, c]], co[[a]]] + Met4DD0[[a, c]] * D[kvf4[[i, c]], co[[b]]], {c, 4}], {i, 3}, {a, 4}, {b, 4}]]
```

```
Out[99]=
```

$$\{\{\{0, 0, 0, 0\}, \{0, 0, 0, 0\}\}, \{\{0, 0, 0, 0\}, \{0, 0, 0, 0\}\}, \{\{0, 0, 0, 0\}, \{0, 0, 0, 0\}\}\}, \{\{\{0, 0, 0, 0\}, \{0, 0, 0, 0\}\}, \{\{0, 0, 0, 0\}, \{0, 0, 0, 0\}\}, \{\{0, 0, 0, 0\}, \{0, 0, 0, 0\}\}\}$$

Transformation of perturbation

Finally, we calculate how the metric perturbations transform under rotations.

```
In[100]:=
rotpert = tensimp[Table[Sum[kvf4[[i, c]] * D[Met4DD1[[a, b]], co[[c]]] + Met4DD1[[c, b]] * D[kvf4[[i, c]], co[[a]]] + Met4DD1[[a, c]] * D[kvf4[[i, c]], co[[b]]], {c, 4}], {i, 3}, {a, 4}, {b, 4}] /. yru];
```

We see that rotation around the z axis simply gives us the magnetic quantum number m.

```
In[101]:=
TableForm[tensimp[rotpert[[3]] + I m Met4DD1]]
```

```
Out[101]//TableForm=
  0   0   0   0
  0   0   0   0
  0   0   0   0
  0   0   0   0
```

Rotation around the other axes corresponds to the ladder operators. Hence, they allow us to change the value of m.

```
In[102]:=
TableForm[tensimp[rotpert[[1]] + rotpert[[2]] I - (Met4DD1 /. {Y -> (Exp[I #4] I (Derivative[0, 0, 1, 0][Y][#1, #2 - 1, #3, #4] Sqrt[1 - #3 ^ 2] + Y[#1, #2 - 1, #3, #4] (#2 - 1) #3 / Sqrt[1 - #3 ^ 2]) &})] /. yru /. {m -> m + 1}]]]
```

```
Out[102]//TableForm=
  0   0   0   0
  0   0   0   0
  0   0   0   0
  0   0   0   0
```

```
In[103]:=
TableForm[tensimp[rotpert[[1]] - rotpert[[2]] I - (Met4DD1 /. {Y -> (Exp[-I #4] I (-Derivative[0, 0, 1, 0][Y][#1, #2 + 1, #3, #4] Sqrt[1 - #3 ^ 2] + Y[#1, #2 + 1, #3, #4] (#2 + 1) #3 / Sqrt[1 - #3 ^ 2]) &})] /. yru /. {m -> m - 1}]]]
```

```
Out[103]//TableForm=
  0   0   0   0
  0   0   0   0
  0   0   0   0
  0   0   0   0
```

Regge-Wheeler gauge

Gauge-transforming vector field

We can simplify the following calculations by assuming a particular gauge. To arrive at this gauge, we first consider a general vector field, at the same order as the metric perturbations, and expand it in spherical harmonics.

```
In[104]:=
TableForm[rgvf = Inverse[Met4DD0].Join[{ξt[t, r], ξr[t, r]} YS[l, m, χ, φ], {ξp[t, r] * YVP[l, m, χ, φ] + ξa[t, r] * YVA[l, m, χ, φ]} / r ^ 2]]
```

```
Out[104]//TableForm=
-e^{-2ν} Y_{lm} ξ_t
e^{-2λ} Y_{lm} ξ_r
- \frac{e^{-2λ-2ν} (-e^{2λ+2ν} r^2 + e^{2λ+2ν} r^2 χ^2) \left( -\frac{i m Y_{lm}}{1-χ^2} + \Pi Y'_{lm} \right)}{r^6}
\frac{i m Y_{lm} \Pi + (1-χ^2) \Xi Y'_{lm}}{r^4 (1-χ^2)}
```

Change of the metric perturbation

The change of the linear metric perturbation is given by the Lie derivative of the background metric with respect to the transforming vector field.

In[105]:=

```
TableForm[pertchg = tensimp[Table[Sum[rgvf[[c]] * D[Met4DD0[[a, b]], co[[c]]] + Met4DD0[[c, b]] * D[rgvf[[c]], co[[a]]] + Met4DD0[[a, c]] * D[rgvf[[c]], co[[b]]], {c, 4}], {a, 4}, {b, 4}] /. yru]
```

Out[105]//TableForm=

$$\begin{array}{cccc}
 2 Y_{lm} \left(-e^{-2\lambda+2\nu} \xi_r v' + \xi_{t,t} \right) & Y_{lm} \left(-2 \xi_t v' + \xi_{t,r} + \xi_{r,t} \right) & \frac{\frac{i m Y_{lm} \Xi, t}{-1+\chi^2} + (r^2 \xi_{t+\Pi, t}) Y'_{lm}}{r^2} & \frac{i m Y_{lm} (r^2 \xi_{t+\Pi, t}) - (-1+\chi^2) \Xi, t Y'_{lm}}{r^2} \\
 Y_{lm} \left(-2 \xi_t v' + \xi_{t,r} + \xi_{r,t} \right) & 2 Y_{lm} \left(-\xi_r \lambda' + \xi_{r,r} \right) & \frac{\frac{i m Y_{lm} (-4 \Xi r \Xi, r)}{-1+\chi^2} + (-4 \Pi + r^3 \xi_r + r \Pi, r) Y'_{lm}}{r^3} & \frac{i m Y_{lm} (-4 \Pi + r^3 \xi_r + r \Pi, r) + (-1+\chi^2) (4 \Xi - r \Xi, r) Y'_{lm}}{r^3} \\
 \frac{\frac{i m Y_{lm} \Xi, t}{-1+\chi^2} + (r^2 \xi_{t+\Pi, t}) Y'_{lm}}{r^2} & \frac{\frac{i m Y_{lm} (-4 \Xi r \Xi, r)}{-1+\chi^2} + (-4 \Pi + r^3 \xi_r + r \Pi, r) Y'_{lm}}{r^3} & \frac{e^{-2\lambda} \left(2 Y_{lm} \left(\theta^{2\lambda} (-i m \chi \Xi + (m^2 + 1)(1+\lambda)(-1+\chi^2)) \Pi \right) - r^3 (-1+\chi^2) \xi_r \right) - 2 \theta^{2\lambda} (-1+\chi^2) (-i m \Xi + \chi \Pi) Y'_{lm}}{r^2 (-1+\chi^2)^2} & \frac{Y_{lm} \left((-2 m^2 - 1)(1+\lambda)(-1+\chi^2) \Xi - 2 i m \chi \Pi \right) + 2 (\chi \Xi + i m \Pi) Y'_{lm}}{-1+\chi^2} \\
 \frac{i m Y_{lm} (r^2 \xi_{t+\Pi, t}) - (-1+\chi^2) \Xi, t Y'_{lm}}{r^2} & \frac{i m Y_{lm} (-4 \Pi + r^3 \xi_r + r \Pi, r) + (-1+\chi^2) (4 \Xi - r \Xi, r) Y'_{lm}}{r^3} & \frac{Y_{lm} \left((-2 m^2 - 1)(1+\lambda)(-1+\chi^2) \Xi - 2 i m \chi \Pi \right) + 2 (\chi \Xi + i m \Pi) Y'_{lm}}{-1+\chi^2} & \frac{2 \left(Y_{lm} (i m \chi \Xi - m^2 \Pi - \theta^{-2\lambda} r^3 (-1+\chi^2) \xi_r) + (-1+\chi^2) (-i m \Xi + \chi \Pi) Y'_{lm} \right)}{r^2}
 \end{array}$$

Transformed perturbation

To study how the different components of the metric change, we now calculate their sum - this is the perturbation of the metric after performing the gauge transformation.

In[106]:=

```
gtmet = tensimp[pertchg + Met4DD1];
```

Decomposition of components

In order to choose a particularly convenient gauge for the further calculations, it is helpful to study how the different components of the perturbation transform. For this purpose, it is helpful to first gather the time and radial components of the gauge transforming covector field into a two-dimensional object.

In[107]:=

```
xitr = {ξt[t, r], ξr[t, r]}
```

Out[107]=

```
{ξt, ξr}
```

The three scalar components transform as a symmetric two-tensor in the time and radial coordinates. Hence, we get their transformation from a symmetrized covariant derivative of the two-dimensional covector field.

In[108]:=

```
gtsca = Union[Flatten[MapThread[Rule, {{{sp1[t, r], sp2[t, r]}, {sp2[t, r], sp3[t, r]}}, -Table[D[xitr[[i]], co[[j]]] + D[xitr[[j]], co[[i]]] - 2 Sum[Gam4UDD[[k, i, j]] * xitr[[k]], {k, 2}], {i, 2}, {j, 2}]/. ε -> 0], 2]]]
```

Out[108]=

```
{Stt -> 2 e-2λ+2ν ξr v' - 2 ξt,t, Str -> 2 ξt v' - ξt,r - ξr,t, Srr -> 2 ξr λ' - 2 ξr,r}
```

The axial vector components transform as the gradient of the axial component of the gauge vector field, properly renormalized with the radius.

In[109]:=

```
gtaxi = MapThread[Rule[#, -D[ξa[t, r]/r^4, #2] r^2] &, {{va1[t, r], va2[t, r]}, {t, r}}, 1]
```

Out[109]=

```
{At -> - $\frac{\Xi, t}{r^2}$ , Ar -> r2  $\left( \frac{4 \Xi}{r^5} - \frac{\Xi, r}{r^4} \right)}$ }
```

Similarly, the polar vector components transform as the gradient of the polar component of the gauge vector field, but with an additional contribution from the time and radial part.

In[110]:=

```
gtvec = MapThread[Rule[#, -#2 - D[ξp[t, r]/r^4, #3] r^2] &, {{vp1[t, r], vp2[t, r]}, xitr, {t, r}}, 1]
```

Out[110]=

$$\left\{ V_t \rightarrow -\xi_t - \frac{\Pi_{,t}}{r^2}, V_r \rightarrow -\xi_r - r^2 \left(-\frac{4\Pi}{r^5} + \frac{\Pi_{,r}}{r^4} \right) \right\}$$

Finally, the tensor components have a simple transformation behavior.

In[111]:=

```
gtten = {tp[t, r] → -2 ξp[t, r]/r^2, ta[t, r] → -2 ξa[t, r]/r^2, tt[t, r] → -2 Exp[-2 λ[r]] ξr[t, r]}
```

Out[111]=

$$\left\{ P \rightarrow -\frac{2\Pi}{r^2}, E \rightarrow -\frac{2\Xi}{r^2}, T \rightarrow -2 e^{-2\lambda} r \xi_r \right\}$$

We check that these indeed give the full transformation.

In[112]:=

```
TableForm[tensimp[gtmet /. gtsca /. gtaxi /. gtvec /. gtten]]
```

Out[112]//TableForm=

$$\begin{array}{cccc} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{array}$$

Gauge choice

From the gauge transformations above, we see that by an appropriate choice of the gauge vector field we can eliminate the two off-diagonal tensor components of the perturbation, as well as the polar vector parts. This is achieved by choosing the following vector field:

In[113]:=

```
rwgauge = {ξp → (-#2^2 tp[#1, #2]/2 &), ξa → (-#2^2 ta[#1, #2]/2 &), ξt → (Derivative[1, 0][tp][#1, #2]/2 - vp1[#1, #2] &), ξr → (Derivative[0, 1][tp][#1, #2]/2 - tp[#1, #2]/#2 - vp2[#1, #2] &)}
```

Out[113]=

$$\left\{ \xi_p \rightarrow \left(\frac{1}{2} (-\#2^2) P \& \right), \xi_a \rightarrow \left(\frac{1}{2} (-\#2^2) E \& \right), \xi_t \rightarrow \left(\frac{P_{,t}}{2} - V_t \& \right), \xi_r \rightarrow \left(\frac{P_{,r}}{2} - \frac{P}{\#2} - V_r \& \right) \right\}$$

We see that indeed now the tensor part becomes diagonal, and the vector part retains only axial components:

In[114]:=

```
TableForm[rwmet = tensimp[gtmet /. rwgauge]]
```

Out[114]//TableForm=

$$\begin{array}{cccc} Y_{lm} \left(S_{tt} + \frac{e^{-2\lambda} v' (2P + 2r V_r - r P_{,r})}{r} - 2V_{t,t} + P_{,tt} \right) & Y_{lm} \left(S_{tr} + 2V_t v' - V_{t,r} - \frac{(1+r v') P_{,t}}{r} - V_{r,t} + P_{,tr} \right) & \frac{i m Y_{lm} (2A_t - E_{,t})}{2(-1+\chi^2)} & -\frac{1}{2} (-1+\chi^2) (2A_t - E_{,t}) Y'_{lm} \\ Y_{lm} \left(S_{tr} + 2V_t v' - V_{t,r} - \frac{(1+r v') P_{,t}}{r} - V_{r,t} + P_{,tr} \right) & \frac{Y_{lm} (2P + r (2P\lambda' - 2P_{,r} + r (S_{rr} + \lambda' (2V_r - P_{,r}) - 2V_{r,r} + P_{,rr})))}{r^2} & \frac{i m Y_{lm} (2E + 2r A_r - r E_{,r})}{2r(-1+\chi^2)} & -\frac{(-1+\chi^2) (2E + 2r A_r - r E_{,r}) Y'_{lm}}{2r} \\ \frac{i m Y_{lm} (2A_t - E_{,t})}{2(-1+\chi^2)} & \frac{i m Y_{lm} (2E + 2r A_r - r E_{,r})}{2r(-1+\chi^2)} & -\frac{e^{-2\lambda} Y_{lm} (-2P + e^{2\lambda} T + r (-2V_r + P_{,r}))}{-1+\chi^2} & 0 \\ -\frac{1}{2} (-1+\chi^2) (2A_t - E_{,t}) Y'_{lm} & -\frac{(-1+\chi^2) (2E + 2r A_r - r E_{,r}) Y'_{lm}}{2r} & 0 & -e^{-2\lambda} (-1+\chi^2) Y_{lm} (-2P + e^{2\lambda} T + r (-2V_r + P_{,r})) \end{array}$$

Hence, we can assume that the metric perturbation is already given in this gauge, which is called the Regge-Wheeler gauge.

In[115]:=

```
rwrw = {ta → (0 &), tp → (0 &), vp1 → (0 &), vp2 → (0 &)};
```

One finds that imposing the Regge-Wheeler gauge on either the transformed or untransformed metric perturbation gives the same result, since it means that the gauge transformation becomes trivial if one already starts from the Regge-Wheeler gauge.

In[116]:=

TableForm[tensimp[Met4DD1 /. rwru]]

Out[116]//TableForm=

$$\begin{array}{cc} S_{tt} Y_{lm} & S_{tr} Y_{lm} \\ S_{tr} Y_{lm} & S_{rr} Y_{lm} \\ \frac{i m A_t Y_{lm}}{-1+\chi^2} & \frac{i m A_r Y_{lm}}{-1+\chi^2} \\ -((-1+\chi^2) A_t Y'_{lm}) & -((-1+\chi^2) A_r Y'_{lm}) \end{array} \quad \begin{array}{cc} \frac{i m A_t Y_{lm}}{-1+\chi^2} & -((-1+\chi^2) A_t Y'_{lm}) \\ \frac{i m A_r Y_{lm}}{-1+\chi^2} & -((-1+\chi^2) A_r Y'_{lm}) \\ -\frac{T Y_{lm}}{-1+\chi^2} & 0 \\ 0 & -((-1+\chi^2) T Y_{lm}) \end{array}$$

In[117]:=

TableForm[tensimp[rwmet /. rwru]]

Out[117]//TableForm=

$$\begin{array}{cc} S_{tt} Y_{lm} & S_{tr} Y_{lm} \\ S_{tr} Y_{lm} & S_{rr} Y_{lm} \\ \frac{i m A_t Y_{lm}}{-1+\chi^2} & \frac{i m A_r Y_{lm}}{-1+\chi^2} \\ -((-1+\chi^2) A_t Y'_{lm}) & -((-1+\chi^2) A_r Y'_{lm}) \end{array} \quad \begin{array}{cc} \frac{i m A_t Y_{lm}}{-1+\chi^2} & -((-1+\chi^2) A_t Y'_{lm}) \\ \frac{i m A_r Y_{lm}}{-1+\chi^2} & -((-1+\chi^2) A_r Y'_{lm}) \\ -\frac{T Y_{lm}}{-1+\chi^2} & 0 \\ 0 & -((-1+\chi^2) T Y_{lm}) \end{array}$$

Ricci tensor perturbation

For later use, we now calculate the Ricci tensor perturbation in the Regge-Wheeler gauge.

In[118]:=

Ric4DD1RW = tensimp[Ric4DD1 /. rwru];

Perturbation of the Schwarzschild metric

Schwarzschild metric

In the following, we will assume that the background solution is given by the Schwarzschild metric.

In[119]:=

sbhru = {v → (Log[1 - R/##]/2 &), λ → (-Log[1 - R/##]/2 &)};

In[120]:=

MatrixForm[Met4DD0 /. sbhru]

Out[120]//MatrixForm=

$$\begin{pmatrix} -1 + \frac{R}{r} & 0 & 0 & 0 \\ 0 & \frac{1}{1-\frac{R}{r}} & 0 & 0 \\ 0 & 0 & \frac{r^2}{1-\chi^2} & 0 \\ 0 & 0 & 0 & r^2(1-\chi^2) \end{pmatrix}$$

Background solution

We check that this is indeed a vacuum solution of Einstein's equations. Of course, this is known - here we just double check.

```
In[121]:= tensimp[Ric4DD0 /. sbhru]
Out[121]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}
```

Ricci tensor perturbation

We then evaluate the perturbation of the Ricci tensor for the Schwarzschild metric.

```
In[122]:= Ric4DD1RWS = tensimp[Ric4DD1RW /. sbhru];
```

Fourier transformation

In the next step, we perform a Fourier transformation with respect to the time coordinate. Here we consider only a single Fourier mode with frequency ω . We thus separate the perturbations into time and radial dependence.

```
In[123]:= fourier = {tt -> (Exp[I \omega #1] ftt[#2] &), sp1 -> (Exp[I \omega #1] fsp1[#2] &), sp2 -> (Exp[I \omega #1] fsp2[#2] &), sp3 -> (Exp[I \omega #1] fsp3[#2] &), va1 -> (Exp[I \omega #1] fva1[#2] &), va2 -> (Exp[I \omega #1] fva2[#2] &)};
```

In the resulting formula for the Ricci tensor perturbation, we find factors $i\omega$ in place of the time derivatives.

```
In[124]:= Ric4DD1RWSF = tensimp[Exp[-I \omega t] Ric4DD1RWS /. fourier];
```

Case $m = 0$

We finally restrict ourselves to the case $m = 0$, since the spherical symmetry implies that the solution will not depend on m .

```
In[125]:= Ric4DD1RWSF0 = tensimp[Ric4DD1RWSF /. {m -> 0}];
```

Decomposing the axial equations

Non-trivial axial equations

Looking at the perturbation of the Ricci tensor, one can see that the axial and polar perturbations decouple. Here we focus on the axial equations. We see that there are three equations, which after canceling some common factors are given as follows.

```
In[126]:= eq03 = FullSimplify[2 r^3 Ric4DD1RWSF0[[1, 4]] / (-1 + x^2) / D[Y[l, 0, x, \phi], x]
Out[126]= -l (1 + l) r \hat{A}_t + 2 R \hat{A}_t + r (r - R) (-i \omega (2 \hat{A}_r + r \hat{A}_r') + r \hat{A}_t'')
```



```
In[127]:=
eq13 = FullSimplify[2 r^2 (r - R) Ric4DD1RWSF0[[2, 4]] / (-1 + x^2) / D[Y[l, 0, x, phi], x]]
```

```
Out[127]=
-2 i r^2 omega A_t + (-((-2 + l + l^2) (r - R)) + r^3 omega^2) A_r + i r^3 omega A_t'
```

```
In[128]:=
eq23 = FullSimplify[2 r^2 (r - R) Ric4DD1RWSF0[[3, 4]] / (l (l + 1) Y[l, 0, x, phi] - 2 D[Y[l, 0, x, phi], x])]
```

```
Out[128]=
i r^3 omega A_t + R (-r + R) A_r - r (r - R)^2 A_r'
```

Relation between equations

Note first that the three equations above are not independent of each other. It turns out that the first equation is a linear combination of the second, the third and the derivative of the second equations. To see this, we gather these four equations into a list.

```
In[129]:=
TableForm[eqnax = {eq03, eq13, eq23, D[eq13, r]]]
```

```
Out[129]//TableForm=
- l (1 + l) r A_t + 2 R A_t + r (r - R) (-i omega (2 A_r + r A_r') + r A_t'')
- 2 i r^2 omega A_t + (-((-2 + l + l^2) (r - R)) + r^3 omega^2) A_r + i r^3 omega A_t'
i r^3 omega A_t + R (-r + R) A_r - r (r - R)^2 A_r'
- 4 i r omega A_t + (2 - l - l^2 + 3 r^2 omega^2) A_r + i r^2 omega A_t' + (-((-2 + l + l^2) (r - R)) + r^3 omega^2) A_r' + i r^3 omega A_t''
```

The independent functions which appear in these equations are the following axial perturbation components and their derivatives.

```
In[130]:=
terms = {fva1[r], fva1'[r], fva1''[r], fva2[r], fva2'[r]}
```

```
Out[130]=
{A_t, A_t', A_t'', A_r, A_r'}
```

Now we have four equations and five terms which appear in them. The coefficients of these five terms are given as follows.

```
In[131]:=
TableForm[coeff = Outer[Coefficient, eqnax, terms]]
```

```
Out[131]//TableForm=
- l (1 + l) r + 2 R      0      r^3 - r^2 R      - 2 i r^2 omega + 2 i r R omega      - i r^3 omega + i r^2 R omega
- 2 i r^2 omega      i r^3 omega      0      -((-2 + l + l^2) (r - R)) + r^3 omega^2      0
i r^3 omega      0      0      R (-r + R)      - r (r - R)^2
- 4 i r omega      i r^2 omega      i r^3 omega      2 - l - l^2 + 3 r^2 omega^2      -((-2 + l + l^2) (r - R)) + r^3 omega^2
```

If the equations are independent, then the matrix above has maximal rank, which is 4. However, we see that this is not the case.

```
In[132]:=
MatrixRank[coeff]
```

```
Out[132]=
3
```

It follows that the transposed matrix has a non-trivial, one-dimensional kernel, which we can find as follows (and renormalize for simplicity).

```
In[133]:=
ns = NullSpace[Transpose[coeff][[1]] r (r - R)
```

```
Out[133]=
{-i r^2 ω, -r + R, 2 - l - l^2, r (r - R)}
```

Finally, we check that the corresponding linear combination of the equations indeed vanishes.

```
In[134]:=
FullSimplify[eqnax.ns]
```

```
Out[134]=
0
```

Solve equation for time component

We are now left with two equations for two unknown functions. Note that the last equation does not contain derivatives acting on the time component of the perturbation. Hence, we may solve it algebraically.

```
In[135]:=
sol23 = {fva1 → (Evaluate[FullSimplify[Solve[eq23 == 0, fva1[r][[1]][[1, 2]] /. {r → #1} &])
```

```
Out[135]=
{fva1 → { - (i (-R + #1) (R \hat{A}_r + #1 (-R + #1) \hat{A}'_r)) / (ω #1^3) &}}
```

Remaining equation

To obtain an equation for the radial component, we insert the solution for the time component into the only remaining equation.

```
In[136]:=
eqax = FullSimplify[eq13 /. sol23]
```

```
Out[136]=
(-((-2 + l + l^2) r^2) + (-6 + l + l^2) r R + 5 R^2 + r^4 ω^2) \hat{A}_r / r + (r - R) ((-2 r + 5 R) \hat{A}'_r + r (r - R) \hat{A}''_r)
```

Simplification using tortoise coordinate

To simplify the last equation, we change the radial coordinate. The coordinate we choose here is called the tortoise coordinate. We use the following transformation.

```
In[137]:=
tortoise = {fva2 → {fax[# + Log[#/R - 1] R] # / (1 - R/#) &}};
```

This is the final equation we obtain. Note that the appearing derivative is with respect to the tortoise coordinate, not with respect to the original radial coordinate.

```
In[138]:=
eqaxt = FullSimplify[(r - R) eqax / r^5 /. tortoise]
```

```
Out[138]=
(-((l (1 + l) r - 3 R) (r - R) + r^4 ω^2) \mathcal{A}) / r^4 + \mathcal{A}''
```